
Abstract

An implementation of packages for Pólya theory and combinatorial species for
the computer algebra system MAXIMA is presented.

1 Introduction

A large part of combinatorics is concerned with counting. Many counting problems
can be efficiently solved by using generating functions. In this paper we describe
implementations of two counting methods which are based on generating func-
tions for the computer algebra system MAXIMA[1].

Pólya theory [2] is an important counting method when some objects that are
beeing counted have to be considered equal because of symmetry. The package
discrete provides basic functionaliy for working with permutation groups and
applications of Pólya’s theorems. The package also adds some extensions to the
maxima graphs package and defines functions for other topics in discrete mathe-
matics, but this functionality will not be described in this paper.

The package Species implements methods for working with combinatorial spe-
cies [3, 4]. The theory of combinatorial species can be applied when we are count-
ing objects which are built from smaller objects using some production rules. The
most well known package for combinatorial species is the combstruct package for
Maple. Open–source implementations include MuPAD-Combinat [6], [7], Aldor-
Combinat [5] and the species package of Sage [8]. The package Species described
in this paper currently only implements unlabelled species only and provides func-
tions for counting, listing and random generation of combinatorial species.

The paper is divided into two parts. The first describes Pólya theory and the
second combinatorial species. Both parts start with short theoretical background
and then expose the packages with worked examples in MAXIMA. The packages
require MAXIMA version 5.22 or above.

2 Pólya theory

2.1 Permutation groups

The package discrete provides several functions for working with permutations.
Some basic group theory functions are also present. The purpose is applications in

Pólya theory.
The function permutation product computes a product of permutations. By

default the product is from left to right. The order can be specified with the op-
tion variable permutation multiplication (valid values are left to right and
right to left. A permutation power can be efficiently computed with the func-
tion permutation power.

The function permutation to cycles transforms a permutation to the product
of disjoint cycles. The reverse can be done with permutation from cycles. There
are similar functions for converting a permutation to a product of transpositions
and back.

(%i1) load(discrete)$

(%i2) a: random_permutation([1,2,3,4,5,6,7,8]);

(%o2) [2,5,8,3,7,4,1,6]

(%i3) permutation_to_cycles(%);

(%o3) [[1,2,5,7],[3,8,6,4]]

(%i4) permutation_from_cycles([[1,2,3,4], [5,6]], 10);

(%o4) [2,3,4,1,6,5,7,8,9,10]

(%i5) permutation_power(a, 10);

(%o5) [5,7,6,8,1,3,2,4]

A permutation group is represented as a set of permutation. It can be generated
from a set of generators with the function group from generators (some functions
accept a set of generators instead of the whole group and an optional argument
generators=true).

(%i6) group_from_generators({[2,3,4,5,1], [1,5,4,3,2]});

(%o6) {[1,2,3,4,5],[1,5,4,3,2],[2,1,5,4,3],[2,3,4,5,1],

[3,2,1,5,4],[3,4,5,1,2],[4,3,2,1,5],[4,5,1,2,3],

[5,1,2,3,4],[5,4,3,2,1]}

An important class of problems deal with groups of automorphisms of graphs. The
package discrete provides a function graph automorphisms which can generate
the group of automorphisms of the graph or only the generators of the group.

(%i7) g:grid_graph(3,7)$

(%i8) graph_automorphisms(g);

(%o8) {[1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,20,21],

[3,2,1,6,5,4,9,8,7,12,11,10,15,14,13,18,17,16,21,20,19],

[19,20,21,16,17,18,13,14,15,10,11,12,7,8,9,4,5,6,1,2,3],

[21,20,19,18,17,16,15,14,13,12,11,10,9,8,7,6,5,4,3,2,1]}

For larger graphs an externall program called bliss [9] can be used to find the
automorphisms. The bliss program must be installed separately and the path to
the program must be specified in the variable bliss program.

(%i9) bt4: binary_tree(4)$

(%i10) gens: graph_automorphisms(bt4,program=bliss,generators=true)$

(%i11) group_order(gens);

(%o11) 32768

(%i12) group_orbits(gens, generators=true);

(%o12) {{1,2,4,5,8,9,11,12,16,17,19,20,23,24,26,27},

{3,6,10,13,18,21,25,28},{7,14,22,29},{15,30},{31}}

2.2 Pólya theory

A cycle index of a permutation π is the monomial Z(π; x1, . . . , xk) = xn1
1 xn2

2 · · · x
nk
k

where ni is the number of cycles of length i in the representation of π as the product
of disjoint cycles.

(%i1) load(discrete)$

(%i2) pi: random_permutation(int_range(10));

(%o2) [4,9,8,3,5,10,7,1,2,6]

(%i3) permutation_to_cycles(pi);

(%o3) [[1,4,3,8],[2,9],[6,10]]

(%i4) cycle_index_permutation(pi);

(%o4) x[1]^2*x[2]^2*x[4]

The cycle index of a permutation group G is the average of cycle indices of permu-
tations in G:

Z(G; x1, . . . , xk) =
1
|G| ∑

π∈G
Z(π; x1, . . . , xk).

There are many functions to compute the cycle index of groups.

(%i5) s4: symmetric_group(4)$

(%i6) cycle_index_group(s4);

(%o6) (6*x[4]+8*x[1]*x[3]+3*x[2]^2+6*x[1]^2*x[2]+x[1]^4)/24

(%i7) cycle_index_symmetric(4);

(%o7) (6*x[4]+8*x[1]*x[3]+3*x[2]^2+6*x[1]^2*x[2]+x[1]^4)/24

(%i8) cycle_index_dihedral(7);

(%o8) (6*x[7]+x[1]^7)/14+(x[1]*x[2]^3)/2

Let X and Y be finite sets and F = { f : X → Y} the set of mappings from X to Y.
Let G be a group acting on X. We define an equivalence relation ∼G on F as

f1 ∼G f2 ⇐⇒ ∃g ∈ G ∀x ∈ X : f1(x) = f2(xg).

We say that f1 and f2 differ by the symmetry g. We are interested in the number of
equivalence classes of the relation ∼G. The equivalence classes are called patterns.

Theorem 1 If the size of the set Y is |Y| = r then the number of equivalence classes in the
relation ∼G is

Z(G; r, . . . , r).

The can be applied with the subst inventory(r, ci) command.
Suppose further we have a set W of weights and a weight function w : Y → W.

For a pattern C and f ∈ C we define a pattern inventory of C as

PI(C) = ∏
x∈X

w( f (x)).

Figure 1: 4–ary necklaces of with 2 colors.

Note that the choice of f is not important.
A pattern inventory of the group G is then defined as

PI(G) = ∑
n1+···+nk=n

τ(n1, . . . , nk)[w(yi)]
n1 · · · [w(yk)]

nk .

where τ(n1, . . . , nk) is the number of patterns with the pattern inventory

[w(yi)]
n1 · · · [w(yk)]

nk .

Theorem 2 Let si = [w(y1)]
i + · · ·+ [w(yr)]i. Then

PI(G) = Z(G; s1, . . . , sk).

The theorem can be applied with the subst inventory([w(y1),...,w(yk)], ci)

command.

2.3 Examples

2.3.1 Necklaces and bracelets

A n-ary necklace of is composed of n colored beads arranged in a circle. Two neck-
laces are the same if we can rotate the first necklace so that the colors of the beads
match the colors of the second necklace.

Let X be the set of n beads arranged on a circle and Y the set of colors. A n-ary
necklace is a mapping f : X → Y which assigns the color f (x) to each bead x ∈ X.
Let G be a cyclic group of order n acting on the set X. Two necklaces f1 and f2 are
the same if there is an element g ∈ G such that f2(x) = f1(xg) for all beads x ∈ X.
In order to count the number of n-ary necklaces we apply Theorem 1. If we we
need to substitute r = |Y| into the cycle index of the group G we obtain the number
of n-ary necklaces.

The number of 4-ary necklaces colored with two or four colors can be computed
in MAXIMA as follows.

Figure 2: 6-ary necklaces colored with three black and three white beads.

(%i1) load(discrete)$

(%i2) ci: cycle_index_cyclic(4);

(%o2) (2*x[4]+x[2]^2+x[1]^4)/4

(%i3) subst_inventory(2, ci);

(%o3) 6

(%i4) subst_inventory(4, ci);

(%o4) 70

The necklaces with colors white and black are shown in Figure 1. Using the Theo-
rem 2 we can further divide the numbers according to how many beads are colored
with the color black. We assing weight 1 to the color white and weight b to the color
black. The coefficient at bm tells us the number of necklaces with m beads colored
with the color black.

(%i5) subst_inventory([1,b], ci);

(%o5) b^4+b^3+2*b^2+b+1

If we are interested in necklaces with four colors in which colors red and green both
appear exactly once, we assing the weights r and g to colors red and green and the
weigths 1 to the other two colors.

When we substitute the inventory into the cycle index we obtain the pattern
inventory, from which we need to read the coefficient at the monomial rg.

(%i6) inv: subst_inventory([r,g,1,1], ci);

(%o6) r^4+g*r^3+2*r^3+2*g^2*r^2+6*g*r^2+7*r^2+g^3*r

+6*g^2*r+12*g*r+8*r+g^4+2*g^3+7*g^2+8*g+6

(%i7) mcoeff(inv,r,1,g,1);

(%o7) 12

If we replace the cyclic group with the dihedral group we also do not distinguish
between necklaces if they differ by a reflection. Such necklaces are called bracelets.

We count the necklaces and bracelets of size 6 with three beads colored black
and three beads colored white.

(%i8) ci: cycle_index_cyclic(6);

(%o8) (2*x[6]+2*x[3]^2+x[2]^3+x[1]^6)/6

(%i9) subst_inventory([1,t], ci);

(%o9) t^6+t^5+3*t^4+4*t^3+3*t^2+t+1

(%i10) coeff(%, t, 3);

(%o10) 4

(%i11) ci: cycle_index_dihedral(6);

Figure 3: A molecular graph.

(%o11) (2*x[6]+2*x[3]^2+x[2]^3+x[1]^6)/12+(x[2]^3+x[1]^2*x[2]^2)/4

(%i12) subst_inventory([1,t], ci);

(%o12) t^6+t^5+3*t^4+3*t^3+3*t^2+t+1

(%i13) coeff(%, t, 3);

(%o14) 3

6-ary necklaces with three black and three white beads are shown in Figure 2. The
second and third necklace represent the same bracelet.

In the last example we show how to obtain the list of 4–ary necklaces with 2
colors using the group obrbits function. We need to define a set of all colorings of
the 4–cycle (represented as a list), and compute orbit representatives for the group
action on the set of colorings.

The possible colorings are binary sequences of length 4 and are computed with
the bin seqs function. We define the function permute list which defines the
action of the group on the colorings. Finally we compute the group orbit repre-
sentatives with the group orbit representatices function. We need to specify
the set on which the group acts and the group action. We also only wish to see
representatives for each orbit. The result should be compared with Figure 1.

(%i15) bin_seqs(n) :=

if n=1 then [[1],[0]]

else block([bs: bin_seqs(n-1)],

append(

map(lambda([s], cons(1, s)), bs),

map(lambda([s], cons(0, s)), bs)))$

(%i16) grp: cyclic_group(4)$

(%i17) group_orbit_representatives(grp,

set=setify(bin_seqs(4)), action=permute_list);

(%o17) {[0,0,0,0],[0,0,0,1],[0,0,1,1],[0,1,0,1],[0,1,1,1],

[1,1,1,1]}

(%i18) length(%);

(%o18) 6

2.3.2 Colorings of graphs

An important application of Pólya theory is to molecular graphs (graphs which
represent molecules). An example of such graph is shown in Figure 3.

In the next example we count 2–colorings of the graph in Figure 3 if we consider
two colorings to be equal when they differ by an automoprhism of the graph. We
also count the number of 2-colorings in which the first color appears twice. The
package discrete loads the graphs package so we can use its functions to define
the graph. We need to work around a technical issue first. The graphs created with
the graphs package have vertex ids from 0 to n− 1. We first relabel the vertices so
that the ids are from 1 to n.

(%i1) load(discrete)$

(%i2) g: cycle_graph(14)$

(%i3) g: relabel_graph_vertices(g, min_id=1)$

(%i4) add_edges([[3,12],[5,10]], g)$

(%i5) grp: graph_automorphisms(g);

(%o5) {[1,2,3,4,5,6,7,8,9,10,11,12,13,14],

[7,6,5,4,3,2,1,14,13,12,11,10,9,8],

[8,9,10,11,12,13,14,1,2,3,4,5,6,7],

[14,13,12,11,10,9,8,7,6,5,4,3,2,1]}

(%i6) ci: cycle_index_group(grp);

(%o6) (2*x[2]^7+x[1]^2*x[2]^6+x[1]^14)/4

(%i7) subst_inventory(2, ci);

(%o7) 4224

(%i8) inv: subst_inventory([1,t], ci);

(%o8) t^14+4*t^13+28*t^12+94*t^11+266*t^10+508*t^9+777*t^8

+868*t^7+777*t^6+508*t^5+266*t^4+94*t^3+28*t^2

+4*t+1

(%i9) coeff(inv, t, 2);

(%o9) 28

If we wish to count the number of 2–colorings of the edges of the graph, we need
to compute the group action of grp on the set edges of the graph.

(%i10) edges: fullsetify(edges(g))$

(%i11) grp_e: group_action(grp, edges);

(%o11) {[1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16],

[9,10,7,6,8,4,3,5,1,2,16,15,14,13,12,11],

[11,10,12,13,8,14,15,5,16,2,1,3,4,6,7,9],

[16,2,15,14,5,13,12,8,11,10,9,7,6,4,3,1]}

(%i12) ci_e: cycle_index_group(grp_e);

(%o12) (2*x[2]^8+x[1]^4*x[2]^6+x[1]^16)/4

(%i13) subst_inventory([1,t], ci_e);

(%o13) t^16+5*t^15+37*t^14+147*t^13+482*t^12+1113*t^11

+2059*t^10+2895*t^9+3290*t^8+2895*t^7+2059*t^6

+1113*t^5+482*t^4+147*t^3+37*t^2+5*t+1

Figure 4: Graphs on 4 vertices.

2.3.3 Graphs generating function

The graph generating function is defined as

GFn(t) =

n(n+1)
2

∑
m=0

amtm

where an is the number of graphs on n vertices with m edges. In this example we
compute the function GF4(t).

Two graphs are isomorphic if there exists a permutation of vertices of the first
graph which produces the second graph.

We think of a graph G on n vertices as a coloring of the edges of the complete
graph Kn with two colors black and white. The black edges of Kn are present in the
graph G and the white are not. Two graphs G and H on n vertices are isomorphic
if there exists an automorphism of the graph Kn which changes the coloring of Kn
corresponding to G to the coloring corresponding to H. The group of automor-
phisms of Kn is the symmetric group Sn. We need to compute the action of Sn on
the edges of Kn which are the 2-subsets of the set {1, 2, . . . , n}. We are now ready
to compute a graphs generating function. Let n = 4.

(%i1) load(discrete)$

(%i2) grp_v: symmetric_group(4)$

(%i3) grp_e: group_action(grp_v, powerset({1,2,3,4}, 2))$

(%i4) ci: cycle_index_group(grp_e);

(%o4) (6*x[2]*x[4]+8*x[3]^2+9*x[1]^2*x[2]^2+x[1]^6)/24

(%i5) subst_inventory(2, ci);

(%o5) 11

(%i6) subst_inventory([1,t], ci);

(%o6) t^6+t^5+2*t^4+3*t^3+2*t^2+t+1

The list of graphs on 4 vertices is shown in Figure 4. The coefficient at tm in the
output %o6 gives the number of graphs with m edges on four vertices.

3 Combinatorial species

The package Species has functions for counting, listing and random selection from
(unlabelled) combinatorial species. Informally, a combinatorial species is a class of
objects which are either atomic or are constructed with a production rule from other
combinatorial species.

Atomic objects can be atoms which have size 1 or epsilon objects which have
size 0. Productions rules implemented are disjoint union (Sum), cartesian product
(Prod), sequence (Seq or Sequence), sets with repetition (Multiset or MSet), sets
without repetition (Set) and cycles (Cycle).

A specification for combinatorial species is a list of production rules. We assume
that all symbols which appear in a specification and do not have production rules
are atoms.

For example, the specification

{S = Prod(X, Y), X = Sum(a, b, c), Y = Sum(e, f )}

defines three combinatorial species. The elements a, b, c, e, f are atoms, species
X contains elements a, b and c, species Y contains elements e and f and finally
species S contains PROD(a, e), PROD(a, f ), PROD(b, e), PROD(b, f ), PROD(c, e),
PROD(c, f ).

The size of an element e from a combinatorial species is the sum of sizes of all
atomic objects which appear in e. For production rules Seq, Set, MSet and Cycle

we also define the cardinality, which is the number of elements in the sequence, set,
multiset and cycle.

For example in specification {A = Sum(x, Prod(x, x)), B = Seq(A)} we have a
species A with elements x and (x, x) and a species B of sequences of elements of A.
The element SEQ(x, x, x, PROD(x, x), x, PROD(x, x)) has cardinality 6 and size 8.

There is also a special species production rule calles Function. The rule accepts
one arguments which is a function which returns the number of elements of given
size.

3.1 Counting, listing, random generation

The main functions in the Species package are count species, list species and
select from species. All take three arguments: the species A, specification spec
and the size n. count species returns the number of elements of the species S
defined by spec of size n, list species returns a list of all elements of the species
S of size n and select from species returns a random element from the species S.

Random generation is uniform, if the number of elements of size n is an, then
the probability that a specific element will be generated is 1

an
.

All three functions will remember all partial results computed which can be
used in later computations. In some situations this memory needs to be cleared.
The memory can be cleared with the function reset species memory.

There is also a function nice disp changes elements obtained from production
rules into lists.

(%i1) load(Species)$

(%i2) count_species(S, spec, 2);

(%o2) 6

(%i3) list_species(S, spec, 2);

(%o3) [PROD(a,e),PROD(a,f),PROD(b,e),PROD(b,f),

PROD(c,e),PROD(c,f)]

(%i4) select_from_species(S, spec, 2);

(%o4) PROD(a,e)

(%i5) nice_disp(%);

(%o5) [a,e]

3.2 Generating functions

For a combinatorial species A we define its generating function A(t) as a formal
power series

A(t) =
∞

∑
n=0

antn

where an is the number of elements of size n in the species A.
The function gf equations(spec,t) returns a system of equations which de-

fine the generating functions of species defined by spec. If the system is nice the
function gf solve can then be used to compute the generating functions from equa-
tions, or the function gf express can be used to find an equation for a specific
generating function.

(%i6) spec: [S=Seq(Sum(x, Prod(x,x)))];

(%o6) [S = Seq(Sum(x,Prod(x,x)))]

(%i7) gf_eqs: gf_equations(spec, t);

(%o7) [S(t) = g4323(t)+1,g4323(t) = g4322(t)*S(t),

g4322(t) = x(t)+g4321(t),g4321(t) = x(t)^2,

Epsilon(t) = 1,x(t) = t]

(%i8) alg_eq: gf_express(gf_eqs, S(t));

(%o8) (-t^2-t+1)*S(t)-1

(%i9) gf_solve(gf_eqs);

(%o9) [[S(t) = -1/(t^2+t-1),g4323(t) = -(t^2+t)/(t^2+t-1),

g4322(t) = t^2+t,g4321(t) = t^2,Epsilon(t) = 1,

x(t) = t]]

When the equation for the generating function A(t) = ∑∞
n=0 antn is algebraic the

function algeq to rec (based on the MuPAD–Combinat function algeqtorec) can
be used to compute a recurrence relation for the sequence an. This can be used to
define a function for fast species counting.

(%i10) rec: algeq_to_rec(alg_eq, S(t));

(%o10) y[n]-y[n-1]-y[n-2]

(%i11) init: makelist(count_species(S, spec, i), i, 1, 2);

Figure 5: Decomposition of a binary tree into smaller binary trees.

(%o11) [1,2]

(%i12) rec_to_function(rec, y[n], init);

(%o12) y_rf[n]:=if n <= 2 then part([1,2],n)

else y_rf[n-1]+y_rf[n-2]

(%i13) y_rf[30];

(%o13) 1346269

3.3 Examples

3.3.1 Necklaces

The number necklaces on four vertices with two colors can be computed with the
Cycle production rule.

(%i1) load(Species)$

(%i2) spec:[N=Cycle(Sum(w,b))];

(%o2) [N = Cycle(Sum(w,b))]

(%i3) count_species(N, spec, 4);

(%o3) 6

(%i4) list_species(N, spec, 4);

(%o4) [CYCLE(w,w,w,w),CYCLE(b,w,w,w),CYCLE(b,w,b,w),

CYCLE(b,b,w,w),CYCLE(b,b,b,w),CYCLE(b,b,b,b)]

Compare this with the necklace example in the Pólya theory section.

3.3.2 Trees

A rooted binary tree can be defined as a leaf or an internal node with two binary
subtrees (see Figure 5). Binary trees with n leaves can be counted with the following
specification

(%i1) load(Species)$

(%i2) spec: [T=Sum(x, Prod(T, T))];

(%o2) [T = Sum(x,Prod(T,T))]

(%i3) makelist(count_species(T, spec, i), i, 1, 10);

(%o3) [1,1,2,5,14,42,132,429,1430,4862]

Figure 6: A random binary tree with 5 leaves.

The generating functions can be used to efficiently count the binary trees for large
number of leaves.

(%i4) gf_equations(spec, t);

(%o4) [T(t) = x(t)+g4255(t),g4255(t) = T(t)^2,x(t) = t]

(%i5) gf_express(%, T(t));

(%o5) T(t)^2-T(t)+t

(%i6) algeq_to_rec(%, T(t));

(%o6) n*y[n]+(6-4*n)*y[n-1]

(%i7) rec_to_function(%, y[n], [1,1])$

(%i8) y_rf[100];

(%o8) 227508830794229349661819540395688853956041682601541047340

(%i9) nice_disp(select_from_species(T, spec, 5));

(%o9) [x,[x,[[x,x],x]]]

The random selection corresponds to the tree in Figure 6.
A ternary tree is a rooted tree in which each node has at most three children.

We do not distinguish between two trees if one can be obtained from the other by
permuting the children of some nodes.

Ternary trees can be counted with the following specification.

(%i10) spec: [TT=Prod(x, MSet(TT, max_card=3))]$

(%i11) makelist(count_species(TT, spec, i), i, 1, 15);

(%o11) [1,1,2,4,8,17,39,89,211,507,1238,3057,7639,19241,48865]

The function tree disp nicely prints a tree.

(%i12) tree_disp(tree) := tree_disp1(tree, 1)$

tree_disp1(tree, d) := (

if length(tree)>0 then

printf(true,"~{~a~}\\-- ~a~%",makelist(" ",d),first(tree)),

if length(tree)>1 then

map(lambda([t],tree_disp1(t,d+4)),second(tree)))$

A random ternary tree on 12 nodes (Figure 7) nicely printed with the tree disp

function.

Figure 7: A random ternary tree.

= ε +

Figure 8: Decomposition of a monotonic path into smaller monotonic paths.

(%i13) tree: nice_disp(select_from_species(TT, spec, 12))$

(%i14) tree_disp(tree)$

\-- x

\-- x

\-- x

\-- x

\-- x

\-- x

\-- x

\-- x

\-- x

\-- x

\-- x

\-- x

3.3.3 Monotonic paths

A monotonic path is a path along the edges of a n × n grid which starts at (0, 0),
ends at (n, n) and is always above the diagonal. Monotonic paths can be counted
with the following specification.

Figure 9: A random monotonic path.

(%i1) load(Species)$

(%i2) spec: [MP=Sum(Epsilon, Prod(Up, MP, Right, MP)),

Right=Epsilon]$

(%i3) makelist(count_species(MP, spec, i),i,1,10);

(%o3) [1,2,5,14,42,132,429,1430,4862,16796]

The sequence is similar to number of binary trees. The numbers in the sequence
are known as Catalan numbers Cn = 1

n+1 (
2n
n ).

(%i4) makelist(binomial(2*n,n)/(n+1),n,1,10);

(%o4) [1,2,5,14,42,132,429,1430,4862,16796]

A random monotonic path of length 8 is shown in Figure 9.

(%i5) flatten(nice_disp1(select_from_species(MP, spec, 8)));

(%o5) [Up,Right,Up,Up,Up,Right,Right,Up,Right,Up,Up,Right,Right,

Right,Up,Right]

We can find the explicit formula for the number of monotonic paths of length n
from the generating functions.

(%i6) gf_equations(spec, t)$

(%i7) gf_express(%, MP(t));

(%o7) t*MP(t)^2-MP(t)+1

(%i8) algeq_to_rec(%, MP(t));

(%o8) (n+1)*y[n]+(2-4*n)*y[n-1]

(%i9) solve_rec(%, y[n], y[1]=1);

(%o9) y[n] = 2^(2*n)*gamma(n+1/2)/(sqrt(%pi)*(n+1)!)

3.3.4 Integer partitions

We will use the Function production rule to define a species of (subset of) integers.
We assume that the size of an integer n is n. An integer partition of n is then a
multiset of integers of size n.

(%i1) load(Species)$

(%i2) cnt_ints(n) := if n>0 then 1 else 0$

(%i3) specIP: [IP=MSet(N), N=Function(cnt_ints)]$

(%i4) count_species(IP, specIP, 100);

(%o4) 190569292

(%i5) num_partitions(100);

(%o5) 190569292

(%i6) nice_disp(select_from_species(IP, specIP, 100));

(%o6) [1,1,1,1,1,1,1,1,3,3,3,3,3,3,3,3,7,7,7,15,32]

Of course the counting function can be more complicated. In the following example
we count the number of partitions of 100 into two primes and two squares.

(%i7) cnt_primes(n) := if primep(n) then 1 else 0$

(%i8) cnt_squares(n) := if integerp(sqrt(n)) then 1 else 0$

(%i9) specIP: [IP=Prod(P, P, S, S),

P=Function(cnt_primes),

S=Function(cnt_squares)]$

(%i10) count_species(IP, specIP, 100);

(%o10) 397

(%i11) flatten(nice_disp(select_from_species(IP, specIP, 100)));

(%o11) [13,67,4,16]

We can also count the number of partitions of an integer n into three primes.

(%i12) specIPP: [IP=MSet(P, card=3), P=Function(cnt_primes)]$

(%i13) makelist(count_species(IP, specIPP, i), i, 6, 100);

(%o13) [1,1,1,2,1,2,2,2,1,3,2,4,2,3,2,5,2,5,3,5,3,7,3,7,2,6,3,9,2,8,

4,9,4,10,2,11,3,10,4,12,3,13,4,12,5,15,4,16,3,14,5,17,3,16,4,

16,6,19,3,21,5,20,6,20,2,22,5,21,6,22,5,28,5,24,7,25,4,29,5,

27,8,29,5,33,4,29,9,33,4,35,5,34,7,30,3]

(%i14) nice_disp(select_from_species(IP, specIPP, 301));

(%o14) [13,47,241]

3.3.5 Sequences

A sequence a1a2a3a4 · · · can be represented as Prod(a1, Prod(a2, Prod(a3, · · ·))). This
enables us to count sequences with special properties.

In the next example we investigate the number of words of length n with letters
a, b and c in which each letter appears an even number of times. We define species

AeBeCe, AeBeCo, . . . which represent words in which each letter appears an even
number of times; a, b appear an even number of times and c an odd number of
times . . .

(%i1) load(Species)$

(%i2) spec:[

AeBeCe=Sum(Epsilon,

Prod(A,AoBeCe), Prod(B,AeBoCe), Prod(C,AeBeCo)),

AeBeCo=Sum(Prod(A,AoBeCo), Prod(B,AeBoCo), Prod(C,AeBeCe)),

AeBoCe=Sum(Prod(A,AoBoCe), Prod(B,AeBeCe), Prod(C,AeBoCo)),

AeBoCo=Sum(Prod(A,AoBoCo), Prod(B,AeBeCo), Prod(C,AeBoCe)),

AoBeCe=Sum(Prod(A,AeBeCe), Prod(B,AoBoCe), Prod(C,AoBeCo)),

AoBeCo=Sum(Prod(A,AeBeCo), Prod(B,AoBoCo), Prod(C,AoBeCe)),

AoBoCe=Sum(Prod(A,AeBoCe), Prod(B,AoBeCe), Prod(C,AoBoCo)),

AoBoCo=Sum(Prod(A,AeBoCo), Prod(B,AoBeCo), Prod(C,AoBoCe))

]$

(%i3) makelist(count_species(AeBeCe, spec, i), i, 0, 20);

(%o3) [1,0,3,0,21,0,183,0,1641,0,14763,0,132861,0,1195743,

0,10761681,0,96855123,0,871696101]

(%i4) makelist(count_species(AeBoCo, spec, i), i, 0, 20);

(%o4) [0,0,2,0,20,0,182,0,1640,0,14762,0,132860,0,1195742,

0,10761680,0,96855122,0,871696100]

(%i5) makelist(count_species(AoBoCo, spec, i), i, 0, 20);

(%o5) [0,0,0,6,0,60,0,546,0,4920,0,44286,0,398580,0,

3587226,0,32285040,0,290565366,0]

(%i6) makelist(count_species(AoBeCe, spec, i), i, 0, 20);

(%o6) [0,1,0,7,0,61,0,547,0,4921,0,44287,0,398581,0,

3587227,0,32285041,0,290565367,0]

We see that for odd n there are no sequences in AeBeCe and for even n there is one
more sequence in AeBeCe than in AeBoCo. We can further investigate the generat-
ing functions.

(%i7) eqns: gf_equations(spec, t)$

(%i8) eqns_sol: gf_solve(eqns)$

(%i9) AeBeCe_gf: assoc(AeBeCe(t), first(eqns_sol));

(%o9) -(7*t^2-1)/(9*t^4-10*t^2+1)

(%i10) AeBeCe_rec: ratfun_to_rec(AeBeCe_gf);

(%o10) -y[n]+10*y[n-2]-9*y[n-4]

(%i11) AeBeCe_sol: solve_rec(AeBeCe_rec, y[n],

y[0]=1, y[1]=0, y[2]=3, y[3]=0);

(%o11) y[n] = 3^n/8+3*(-1)^n/8+(-3)^n/8+3/8

(%i12) declare(n, integer)$

(%i13) subst(n=2*n,AeBeCe_sol);

(%o13) y[2*n] = 3^(2*n)/4+3/4

Figure 10: Blocks used for building towers.

Figure 11: Towers species of size n.

Figure 12: Towers of size 2.

Figure 13: A random tower of size 10.

3.3.6 Building towers

We count the number of different ways to build a two column tower with blocks
shown in Figure 10. We name the block with letters S, V, H, L1, L2, L3 and L4 from
left to right, up to bottom. For blocks S and V we addd the number of the column
in which the block sits.

If we draw a line at the height n we will have three posibilities (Figure 11). The
line does not intersect any block (species LL), it intersects a block V or L2 in the
second column and no block in the first column (species LP) or it intersects a block
V or L1 in the first column and no block in the second column (species PL). Note
that if the line intersects L3 in the first column and no block in the second column
this belongs to the species LL of size n + 1. In the specification there is a symbol x
for the size of the tower. We also have symbols for blocks which are of size 0. Here
is a specification for the species.

(%i1) load(Species)$

(%i2) spec: [

LL=Sum(Epsilon,

Prod(x, LL, S1, S2),

Prod(x, LL, H),

Prod(x,x, LL, V1, V2),

Prod(x, LP, S1),

Prod(x, PL, S2),

Prod(x,x, LP, L3),

Prod(x,x, LL, S1, L4),

Prod(x,x, LL, L3, S2),

Prod(x,x, PL, L4)),

LP=Sum(

Prod(x, LL, L2),

Prod(x, PL, V2),

Prod(x, LL, S1, V2)),

PL=Sum(

Prod(x, LL, L1),

Prod(x, LP, V1),

Prod(x, LL, V1, S2)),

S1=Epsilon, S2=Epsilon, V1=Epsilon, V2=Epsilon, H=Epsilon,

L1=Epsilon, L2=Epsilon, L3=Epsilon, L4=Epsilon]$

We can list the towers of size 2 (shown in Figure 12).

(%i3) list_species(LL, spec, 2)$

(%i4) map(lambda([f], rest(flatten(f), 3)), nice_disp(%));

(%o4) [[S1,S2,S1,S2],[H,S1,S2],[S1,S2,H],[H,H],[V1,V2],

[L2,S1],[S1,V2,S1],[L1,S2],[V1,S2,S2],[S1,L4],

[L3,S2]]

Or choose a random tower of size 10 (shown in Figure 13).

(%i5) select_from_species(LL, spec, 10)$

(%i6) rest(flatten(nice_disp(%)), 11);

(%o6) [S1,S2,V1,S2,S2,V1,S2,L4,H,H,L1,S2]

Using the generating functions we can find an explicit formula for the number of
towers of size n. Since the system is complicated we use Gröbner bases to express
LL(t).

(%i7) makelist(count_species(LL, spec, i), i, 0, 5);

(%o7) [1,2,11,44,189,798]

(%i8) eqs: gf_equations(spec, t)$

(%i9) LL_eq: gf_express(eqs, LL(t), use_grobner=true);

(%o9) (-t^3-5*t^2-3*t+1)*LL(t)+t-1

(%i10) LL_rec: algeq_to_rec(LL_eq, LL(t));

(%o10) y[n]-3*y[n-1]-5*y[n-2]-y[n-3]

(%i11) LL_explicit: solve_rec(LL_rec,y[n],y[0]=1,y[1]=2,y[2]=11);

(%o11) y[n] = (-1)^n/2+(sqrt(5)+2)^n*(3*sqrt(5)+5)/20

-(2-sqrt(5))^n*(3*sqrt(5)-5)/20

which is

yn =
(−1)n

2
+

(√
5 + 2

)n (
3
√

5 + 5
)

20
−

(
2−
√

5
)n (

3
√

5− 5
)

20
.

